Utilidad EVALUA
1 Descripción general
Esta herramienta, examina cada uno de los programas relacionados en el fichero de programas de entrada para obtener un fichero que contenga cada una de las líneas en las que figura, al menos una de las cadenas especificadas en el panel de entrada, y el nombre del párrafo en el que se encuentra dicha línea.
El formato de registro de este fichero de entrada, que proviene de una tabla Word en la que al convertirla a texto se ha utilizado el delimitador “@”, es:

Descripción_programa @ nombre_programa @ Nombre_transaccion

	EDIT       ROOT.USRJMP.FILEPGMS                          Columns 00001 00072 

 Command ===>                                                  Scroll ===> CSR  

 ****** ***************************** Top of Data ******************************
 ==MSG> -Warning- The UNDO command is not available until you change            

 ==MSG>           your edit profile using the command RECOVERY ON.              

 000001 Descripcion Programa              @extracción de progrs. a tratar @TRN  

 000002 CONS. EXPTES POR ETAPA DE UN CENTRO, CONTROL DE CENTROS.@JXD0399  @JXMC 

 000003                                                         @no tratar @    

 000004                                                         @no tratar @    

 . . .
 000010 ENLACE CON XF PARA OBTENER LOS INDICES DE REFERENCIA.   @JXD1231@JXN1   

 000011 CAPTURA CENTRO NUEVO                                    @JXD2060@JXK9   

 000012 BAJA VINCULACION AVALES Y ENTREGAS A CUENTA             @JXD2642@JXBF   

 000013                                                         @no tratar @    

 000014 VALIDA EL EXPEDIENTE DE LOS BIENES A INCORPORAR         @JXE0003@JX05   

 000015                                                         @no tratar @    

 000016                                                         @no tratar @    

 000017 PROGRAMA VALIDACION DE LAS FINCAS DE UN EXPEDIENTE      @JXE0009@JX05   




El Panel de entrada a la utilidad es el siguiente:
	------------------------------------------------------------------------------ 

      JMP DYD2                Extraer registros            USUARIO: USRJMP      

 ------------------------------------------------------------------------------ 

 Command ==>  ________________________________________________________________  

     Este procedimiento examina cada uno de los fuentes relacionados en         

     el fichero de programas de entrada que se especifique.                     

     El formato de este fichero es DescripcionPgm @ NombrePgm @ NombreTRN       

     De cada uno de los fuentes se extrae:                                      

     - todas las líneas que contengan cada una de las cadenas especificadas     

     - y el nombre del párrafo en el que se encuentra dicha línea.              

      Fichero de Programas: ROOT.USRJMP.FILEPGMS_______________________          

      Editar fichero Pgms.: S_  (s/SI)                                          

      Librería de fuentes.: ENVDES02.LIBRGEN.FTES_____________________          

      View de Lib Fuentes.: N_  (s/SI)                                          

      Cadenas a buscar ...: W-CNT-PAN__   ____________   ____________          

                            ____________   ____________   ____________          

      Fichero de salida...: ROOT.USRJMP.SALIDA________________________          

      Editar fichero final: S_  (s/SI)                                          

      Debug: NO  (s/S1/SI)     Numero de programas a tratar : ______     

       


Y, el resultado obtenido es el siguiente:
Cada grupo de registros correspondientes a un programa, viene delimitado por una línea de asteriscos.

Solo aparecen los grupos de registros de aquellos programas que se encuentran en la librería de programas especificada. 

Cada grupo cuenta con la siguiente estructura de grupo:

· Transacción y Nombre de la transacción

· Programa y Nombre del programa

· Líneas de detalle.

	 EDIT       ROOT.USRJMP.SALIDA                              Columns 00001 00072 

 Command ===>                                                  Scroll ===> CSR  

 ****** ***************************** Top of Data ******************************

 ==MSG> -Warning- The UNDO command is not available until you change            

 ==MSG>           your edit profile using the command RECOVERY ON.              

 000001 ***************************                                             

 000002 @TRN@ JXHL            @@ DES_TRN                                        

 000003 @PGM@ JXD0918 @@ OBTENCIóN DE LA FASE SELECCIONADA                      

 000004 PROCESO.                                                                

 000005       MOVE W-CNT-PAN            TO JXZ452-CNT-GES-CD                   

 000006 ***************************                                             

 000007 @TRN@ JXN4            @@ DES_TRN                                        

 000008 @PGM@ JXD1228 @@ ENLACE CON IND. REF.                                   

 000009 PROCESO.                                                                

 000010      MOVE W-CNT-PAN                TO ADA-CNT-CD                       

 000011 ***************************                                             

 000012 @TRN@ JXN1            @@ DES_TRN                                        

 000013 @PGM@ JXD1231 @@ ENLACE CON XF PARA OBTENER LOS INDICES DE REFERENCIA.  

 000014 PROCESO.                                                                

 000015      MOVE W-CNT-PAN                TO AUS-CNT-CD                       

 000016 ***************************                                             

 000017 @TRN@ JX05            @@ DES_TRN                                        




Los elementos utilizados para la ejecución de esta utilidad son:

- ROOT.USRJMP.CMD(EVALUA1) : Fichero de comandos escrito en REXX

- ROOT.USRJMP.PNL(EVALUA1) : Definición del Panel de entrada 
2 Fichero de comandos : ROOT.USRJMP.CMD(EVALUA1)
	/***************************  REXX  ********************************/

/*******    evaluacion de modificaciones                    ********/

/*******************************************************************/

/*                                                                 */

/* dada una lista de programas, este evaluador:                    */

/*   - accedera a cada uno de los fuentes                          */

/*   - memorizará cada parrafo                                     */

/*   - cuando encuentre una de las cadenas buscada                 */

/*            escribirá el parrafo donde se encuentra              */

/*            y la linea que contiene la cadena                    */

/*                                                                 */

/*******************************************************************/

/* parametros que se necesitan:                                    */

/* - fichero que contiene la relacion de programas a examinar      */

/*   filepgm: ROOT.USRJMP.salida.fb80                              */

/* --------------------------------------------------------------- */

/*-----------------------------------------------------------------*/

/*       leer fichero de programas                                 */

/*-----------------------------------------------------------------*/

   panel     = "evalua2"

   filepgm   = "ROOT.USRJMP.FILEPGMS"

   FileOut   = 'ROOT.USRJMP.SALIDA'

   EditOut   = 'S'

   EditPgms  = 'S'

   w_reg     = 0

   w_reg_max = 999999

/*--------------------------------------------------------------------*/

/* en principio se va a suponer que todos los fuentes estan en ENVDES     */

/*--------------------------------------------------------------------*/

/* el siguiente parrafo memoriza el directorio de la libreria de ENVDES   */

/* lista los miembros de un pds                                       */

/* Las 6 primeras lineas contienen informacion del PDS                */

/* a partir de la septima linea se encuentran listados los miembros   */

/**********************************************************************/

   LibPgms   = 'ENVDES02.LIBRGEN.FTES'

   y = outtrap('miembro.')

   address tso "listd ("LibPgms") members"

   y = outtrap('off')

/*--------------------------------------------------------------------*/

/*  Recuperar valores memorizados                                     */

/*--------------------------------------------------------------------*/

   ADDRESS ISPEXEC "VGET maquina   SHARED"

   ADDRESS ISPEXEC "VGET FileOut   SHARED"

   ADDRESS ISPEXEC "VGET EditPgms  SHARED"

   ADDRESS ISPEXEC "VGET EditOut   SHARED"

   ADDRESS ISPEXEC "VGET debug     SHARED"

   ADDRESS ISPEXEC "VGET tratar    SHARED"

   ADDRESS ISPEXEC "VGET (cadena1 cadena2 cadena3) SHared"

   ADDRESS ISPEXEC "VGET (cadena4 cadena5 cadena6) SHared"

/*--------------------------------------------------------------------*/

/*  pedir parametros por pantalla                                     */

/*--------------------------------------------------------------------*/

   if strip(debug) = '' then debug = 'NO'

   if debug  = 'SI' then say 'PDS de fuentes: ' LibPgms

   if debug  = 'SI' then say 'fichero con lista de programas: ' filepgm

   root = ROOT
   user = userid()

   if user    = USRJMP1 Then user = USRJMP
   if FileOut  = '' then FileOut  = root"."user".evalua"

   if EditOut  = '' then EditOut  = 'S'

   if EditPgms = '' then EditPgms = 'S'

do forever

   address ispexec " display panel("panel")"

/** ----------------------- salir con PF3 ----------------------------*/

   rc_panel = rc

   if rc_panel = 8 then leave

/** --------------- examinar si existe el fichero de entrada --------**/

   upper filePgm

   fichero = filepgm

   call existe_fichero
   if result = 98 then do

      say "No existe el fichero :" fichero

      leave

   end

/** ----------------------- editar fichero de programas --------------*/

   upper EditPgms

   if EditPgms = 'E' then EditPgms = 'S'

   if EditPgms = 'SI' then EditPgms = 'S'

   if EditPgms = 'S' then do

      "ispexec edit dataset('"FilePgm"')"
      iterate

   end

/** --------------- examinar si existe la libreria de fuentes -------**/

   upper LibPgms

   fichero = LibPgms

   call existe_fichero
   if result = 98 then do

      say "No existe la Libreria:" libPgms

      leave

   end

/** ----------------------- View de la libreria de Pgms --------------*/

   upper ViewPgms

   if ViewPgms = 'E' then ViewPgms = 'SI'

   if ViewPgms = 'S' then ViewPgms = 'SI'

   if ViewPgms = 'SI' then do

      "ispexec view dataset('"LibPgms"')"
      iterate

   end

/** ----------------------- validar cadenas de entrada  --------------*/

   if strip(cadena1) = '' ,

    & strip(cadena2) = '' ,

    & strip(cadena3) = '' ,

    & strip(cadena4) = '' ,

    & strip(cadena5) = '' ,

    & strip(cadena6) = '' then do

      say 'debe especificar al menos una cadena'

      iterate

   end

/** --------------- leer el fichero de entrada ----------------------**/

   "alloc da("filepgm") f(entrada) shr reuse"

   "execio * diskr entrada (finis stem registro."

   w_num_pgms = registro.0

   if strip(tratar) = ""  then Tratar = 10

   upper debug

   if debug = "D"  then debug = "S"

   if debug = "SI" then debug = "S"

   if debug = "S" ,

    | debug = "S1"  then w_num_pgms = Tratar

   do i=1 to w_num_pgms

      /* ---------------- generar separador --------------- */

      /* trace i */

      if (debug = "S") | (debug = "S1") then do

         say '****************************************************'

         say i ':' registro.i

         say '****************************************************'

      end

      /* ---------------- desglosar entrada ------opcion 1- */

      /*

      if debug = "S" then do

         say registro.i

         /* say Pos( 'abra', 'abracadabra' ) -- shows: 1       */

         del1 = Pos( '@', registro.i)

         /* say Pos( 'abra', 'abracadabra', 2 ) -- shows: 8    */

         del2 = Pos( '@', registro.i,del1 + 1)

         /* say Substr( 'abracadabra', 5 ) -- shows: cadabra   */

         pgm_1 = substr( registro.i,del1 + 1,del2 - del1 - 1)

         trn_1 = substr( registro.i,del2 + 1)

         if debug  = 'S' then say    del1 ' ' del2 ' ' pgm_1 ' 'trn_1

      end

      */

      del1 = Pos( '@', registro.i)

      del2 = Pos( '@', registro.i,del1 + 1)

      lenP = del2 - del1

      /* ---------------- desglosar entrada ------opcion 2- */

      /* str = "abc.def.ghi.JKLMN"                          */

      /* parse var string word1 "." word2 "." word3 "."     */

      parse var registro.i  des_pgm "@" pgm_name "@" trn_name

      if debug  = 'S1' then say lenp ' ' pgm_name ' ' trn_name

      /*----------------------------------------------------*/

      /*--- tratamiento individual -------------------------*/

      /*----------------------------------------------------*/

      if lenp <= 8 then do

         /*----------------------------------------------------*/

         /* examinar si existe un miembro con ese nombre ------*/

         /* en la tabla de miembros del pds de fuentes   ------*/

         /*----------------------------------------------------*/

         m = 6

         do while m < miembro.0

             m = m + 1

             miembro.m = strip(miembro.m)

             if strip(pgm_name) =  miembro.m then do

                queue '***************************'

                queue '@TRN@' trn_name '@@' des_trn

                queue '@PGM@' pgm_name '@@' des_pgm

                call tratar_pgm pgm_name

             end

         end

      end

      /*----------------------------------------------------*/

   end

   "FREE fi(entrada)"

/** --------------- una vez tratados todos los ficheros -------------**/

/** --------------- pasamos el contenido de la cola a fichero -------**/

   call allocar_file_salida
   call pasar_cola_a_fichero
/** --------------- examinar el resultado final ---------------------**/

   upper EditOut

   if EditOut  = 'E' then EditOut  = 'S'

   if EditOut  = 'SI' then EditOut  = 'S'

   if EditOut = 'S' then "ispexec edit dataset('"FileOut"')"

/** --------------- salir del bucle ---------------------------------**/

End
exit

/**********************************************************************/

/**  RUTINAS       RUTINAS       RUTINAS       RUTINAS       RUTINAS **/

/**         RUTINAS       RUTINAS       RUTINAS       RUTINAS        **/

/**********************************************************************/

/*********************************************************************/

existe_fichero:

/*-------------------------------------------------------------------*/

     x = outtrap('On')

     address tso "listd " fichero

     x = outtrap('off')

     if rc = 12 | rc = 20 then do

        tipmsg   = ' de error  '||'/'

        mensaje1 = prog_arq     ||'/'

        mensaje2 = ' no existe en host el fichero:    '||'/'

        mensaje3 = fichero     ||'/'

        mensaje4 = ' por favor, introduzca el nombre correcto.'

        call error_parcial
        return 98

     end

return

/**********************************************************************/

error_parcial:

/*-------------------------------------------------------------------*/

/* En principio siempre va a entrar por la 1  parte del IF  ----------*/

     /* porque ENT_INVOC es igual a ONLINE    ------------------------*/

     IF ENT_INVOC = 'ONLINE' THEN DO

        CALL RUTERRGO TIPMSG MENSAJE1 MENSAJE2 MENSAJE3 MENSAJE4

     END

     ELSE DO

        CALL RUTERRGB TIPMSG MENSAJE1 MENSAJE2 MENSAJE3 MENSAJE4

        ZISPFRC = 23

        ADDRESS ISPEXEC "VPUT ZISPFRC SHARED"

     END

     /* Diferencia con la rutina de ERROR_GRAL:           -----------*/

     /* RETURN y no EXIT 23 para que continue en la REXX  -----------*/

RETURN

/**********************************************************************/

tratar_pgm:

/*--------------------------------------------------------------------*/

/** --------------- el venir aqui es porque existe un miembro -------**/

   fichero = LibPgms'('pgm_name')'

   w_parrafo_memo = ''

/** --------------- leer el fichero de entrada ----------------------**/

   "alloc da("fichero") f(programa) shr reuse"

   "execio * diskr programa (finis stem linea."

   w_num_lins = linea.0

   w_logica = "NO"

   do l=1 to w_num_lins

      /* ---------------- tratar cada linea --------------- */

      /*

      say linea.l

      */

      /*-------- determinar area de logica -----------------*/

      if Pos( 'PROCEDURE', linea.l ) > 0 then do

         w_logica = "SI"

      end

      if w_logica = "SI" then do

         w_linea = linea.l

         call tratar_linea w_linea

      end

   end

return

/**********************************************************************/

tratar_linea:

/*--------------------------------------------------------------------*/

   w_linea_aux = strip(w_linea)

   w_linea_72  = substr(w_linea,1,72)

   w_linea_66  = substr(w_linea,7,66)

   /*-----------------------eliminar comentarios ---------------------*/

   parse var w_linea_aux '*' w_comentario

   if w_comentario ¬= '' then return

   /*-----------------------determinar parrafos  ---------------------*/

   W_area_a = substr(w_linea_72 , 8,4)

   if strip(w_area_a) ¬= '' then do

      w_parrafo = strip(w_linea_66)

      if debug = "S" ,

       | debug = "S1" then say 'tl ' w_parrafo

   end

   /*----------------------- examinar cadenas ------------------------*/

   w_escribir = "NO"

   if strip(cadena1) ¬= '' then do

      if pos( strip(cadena1), w_linea ) > 0 then w_escribir = "SI"

   end

   if strip(cadena2) ¬= '' then do

      if pos( strip(cadena2), w_linea ) > 0 then w_escribir = "SI"

   end

   if strip(cadena3) ¬= '' then do

      if pos( strip(cadena3), w_linea ) > 0 then w_escribir = "SI"

   end

   if strip(cadena4) ¬= '' then do

      if pos( strip(cadena4), w_linea ) > 0 then w_escribir = "SI"

   end

   if strip(cadena5) ¬= '' then do

      if pos( strip(cadena5), w_linea ) > 0 then w_escribir = "SI"

   end

   if strip(cadena6) ¬= '' then do

      if pos( strip(cadena6), w_linea ) > 0 then w_escribir = "SI"

   end

   if w_escribir = "SI" then do

      call escribir_linea
   end

return

/**********************************************************************/

escribir_linea:

/*--------------------------------------------------------------------*/

   if w_parrafo_memo ¬= w_parrafo then do

      if debug = "S1" then say 'ep ' w_parrafo

      queue w_parrafo

      w_parrafo_memo = w_parrafo

   end

   if debug = "S1" then say 'ep ' w_linea_66

   queue w_linea_66

   /*

   w_reg = w_reg + 1

   if w_reg > w_reg_max then exit

   */

return

/**********************************************************************/

allocar_file_salida:

/*--------------------------------------------------------------------*/

 x = outtrap('Off')

 x = outtrap('On')

 address tso "delete '"FileOut"'"

 x = outtrap('Off')

 ref="tracks space(1,1) lrecl(80) recfm(f b) dsorg(ps) blksize(23440)"

 "alloc da("FileOut") f(salida) mod "ref"     reuse"

return

/**********************************************************************/

pasar_cola_a_fichero:

/*--------------------------------------------------------------------*/

    QUEUE 'JMP-FIN'

    pull linea

    j = 0

    do until length(linea) = 0

      j = j + 1

      jcl.j = linea

      if debug = "S1" then do

         say j ":" linea

      end

      pull linea

      upper linea

      IF LINEA = 'JMP-FIN' THEN LEAVE

    end

    /*

    say 'pulsar intro para seguir'

    pull seguir

    */

    /* escritura de la tabla en el fichero de salida */

    "EXECIO * DISKW SALIDA (FINIS STEM jcl."

    "FREE fi(SALIDA)"

return




3 Panel de entrada : ROOT.USRJMP.PNL(EVALUA1) 
	)ATTR DEFAULT(%~!)

    | TYPE(OUJXUT) INTENS(LOW) CAPS(OFF) JUST(RIGHT) PAD(' ')

    } TYPE(TEXT)   INTENS(LOW) CAPS(OFF)

    [ TYPE(TEXT)   COLOR(YELLOW) CAPS(OFF)

    ¬ TYPE(TEXT)   COLOR(red) CAPS(OFF)

    # TYPE(ouJXut) COLOR(red) CAPS(OFF)

    ] TYPE(TEXT)   COLOR(GREEN) CAPS(OFF)

    _ TYPE(INPUT)  INTENS(HIGH) PAD('_')

    { TYPE(input)  iNTENS(high) CAPS(On) just(left)  pad('_')

)BODY

~

%------------------------------------------------------------------------------

~     JMP#maquina~           [Extraer registros~           USUARIO:]&ZUSER  ~

%------------------------------------------------------------------------------

%Command ==> _ZCMD                                                            ~

~

~    Este procedimiento examina cada uno de los fuentes relacionados en

~    el]fichero de programas~de entrada que se especifique.

~    El formato de este fichero es[DescripcionPgm¬@[NombrePgm¬@[NombreTRN

~    ]Fichero de Programas:_FilePgm                                   ~

~    ]Editar fichero Pgms.:_z ~[(s/SI)~

~    ]Libreria de fuentes.:_libPgms                                   ~

~    ]View de Lib Fuentes.:_z ~[(s/SI)~

~

~    De cada uno de los fuentes se extrae:

~    - todas las lineas que contengan cada una de las cadenas especificadas

~    - y el nombre del parrafo en el que se encuentra dicha linea.

~    ]Cadenas a buscar ...:_cadena1     ~ _cadena2     ~ _cadena3     ~

~                          _cadena4     ~ _cadena5     ~ _cadena6     ~

~    ]Fichero de salida...:_FileOut                                   ~

~    ]Editar fichero final:_z ~[(s/SI)~

~

~     Debug:_z ~[(s/S1/SI)~   ]Numero de programas a tratar :_tratar~

~

)INIT

 .ZVARS  = '(EditPgms ,viewPgms,EditOut,debug)'

 &submitir = NO

 .cursor= FilePgm

)PROC

 VER(&libPgms,dsname)

 VER(&filepgm,dsname)

 VER(&FileOut,dsname)

 vput(maquina,LibPgms,FilePgm,FileOut,EditPgms ,editOut,debug)

 vput(cadena1,cadena2,cadena3,cadena4,cadena5,cadena6)

)END







4 Panel de entrada : ROOT.USRJMP.PNL(EVALUA2) 
	)ATTR DEFAULT(%~!)

    | TYPE(OUTPUT) INTENS(LOW) CAPS(OFF) JUST(RIGHT) PAD(' ')

    } TYPE(TEXT)   INTENS(LOW) CAPS(OFF)

    [ TYPE(TEXT)   COLOR(YELLOW) CAPS(OFF)

    ¬ TYPE(TEXT)   COLOR(red) CAPS(OFF)

    # TYPE(output) COLOR(red) CAPS(OFF)

    ] TYPE(TEXT)   COLOR(GREEN) CAPS(OFF)

    _ TYPE(INPUT)  INTENS(HIGH) PAD('_')

    { TYPE(input)  iNTENS(high) CAPS(On) just(left)  pad('_')

)BODY

~

%------------------------------------------------------------------------------

~     JMP#maquina~           [Extraer registros~           USUARIO:]&ZUSER  ~

%------------------------------------------------------------------------------

%Command ==> _ZCMD                                                            ~

~

~    Este procedimiento examina cada uno de los fuentes relacionados en

~    el]fichero de programas~de entrada que se especifique.

~    El formato de este fichero es[DescripcionPgm¬@[NombrePgm¬@[NombreTRN

~    De cada uno de los fuentes se extrae:

~    - todas las lineas que contengan cada una de las cadenas especificadas

~    - y el nombre del parrafo en el que se encuentra dicha linea.

~

~    ]Fichero de Programas:_FilePgm                                   ~

~    ]Editar fichero Pgms.:_z ~[(s/SI)~

~    ]Libreria de fuentes.:_libPgms                                   ~

~    ]View de Lib Fuentes.:_z ~[(s/SI)~

~    ]Cadenas a buscar ...:_cadena1     ~ _cadena2     ~ _cadena3     ~

~                          _cadena4     ~ _cadena5     ~ _cadena6     ~

~    ]Fichero de salida...:_FileOut                                   ~

~    ]Editar fichero final:_z ~[(s/SI)~

~

~     Debug:_z ~[(s/S1/SI)~   ]Numero de programas a tratar :_tratar~

~

)INIT

 .ZVARS  = '(editPgms,viewPgms,EditOut,debug)'

 &submitir = NO

 .cursor= FilePgm

)PROC

 VER(&libPgms,dsname)

 VER(&filepgm,dsname)

 VER(&FileOut,dsname)

 vput(maquina,LibPgms,FilePgm,FileOut,editPgms,editOut,debug)

 vput(cadena1,cadena2,cadena3,cadena4,cadena5,cadena6)

)END

